

## **General Disclaimer**

### **One or more of the Following Statements may affect this Document**

- This document has been reproduced from the best copy furnished by the organizational source. It is being released in the interest of making available as much information as possible.
- This document may contain data, which exceeds the sheet parameters. It was furnished in this condition by the organizational source and is the best copy available.
- This document may contain tone-on-tone or color graphs, charts and/or pictures, which have been reproduced in black and white.
- This document is paginated as submitted by the original source.
- Portions of this document are not fully legible due to the historical nature of some of the material. However, it is the best reproduction available from the original submission.

---

# Computer Program for Calculation of Oxygen Uptake

---

B. L. Castle, G. Castle, and J. E. Greenleaf


---

(NASA-TN-78585) COMPUTER PROGRAM FOR  
CALCULATION OF OXYGEN UPTAKE (NASA) 13 p HC  
A02/MF A01 CSCL 06P

N79-21741

Unclas  
G3/51 14730

April 1979



National Aeronautics and  
Space Administration

---

# Computer Program for Calculation of Oxygen Uptake

---

B. L. Castle

G. Castle

J. E. Greenleaf, Ames Research Center, Moffett Field, California



National Aeronautics and  
Space Administration

**Ames Research Center**  
Moffett Field, California 94035

## GENERAL PROGRAM DESCRIPTION

This program, written in Super Basic, can be used to calculate  $O_2$  uptake and  $CO_2$  production in resting and exercising human test subjects. The set of data from which these calculations are made is entered into the program; included are measurements of  $O_2$  and  $CO_2$  in inspired air and expired respiratory gas, and various gas volumes, temperature, and pressure measurements. A flow diagram of the program is presented in Fig. 1.

The program will accommodate readings from two types of  $O_2$  and  $CO_2$  analyzers: indirect readout analyzers (Steps 10.1 and 12.1) which require the use of calibration curves to convert numerical readings of  $O_2$  and  $CO_2$  concentrations into percent, and direct readout analyzers (Steps 10, 11, 12, 13) which read directly in percent. For the indirect readout analyzers, the slopes and intercepts of the calibration curves must be entered as constants (Steps 2, 3, 4).

The program begins with a prompt (Step 5) that allows the operator to either print a set of instructions for running the program, or to go directly to the data input section of the program. The data can be entered "conversationally" through a series of prompts (Steps 6, 6.1, and 6.2), or they can be entered all at once, according to a specified format, without interruption (Steps 7, 8). Expired water vapor pressure is calculated in either Step 6.2A or 8.1, depending on which option for data input is selected.

Direct or indirect measurements of the concentrations of inspired  $CO_2$  and  $O_2$  can be entered in the program (Steps 14-19, excluding 14.1), to be used later in the ventilation calculations. However, if they are not measured, constant values for inspired  $CO_2$  and  $O_2$  concentrations can be used; 0.03% and 20.93%, respectively, are commonly used. These values have been incorporated into the program.

The major ventilation measurements in the  $O_2$  uptake equation are printed out in Step 20; oxygen uptake, expressed in ml/kg/min, is printed in Step 22.2. The basic equations used to calculate these ventilation measurements were taken from Consolazio et al. (1).

A "no" input in response to the query "are you finished?" (Step 23) recycles the program to Step 6 for the next set of input data. One run through the program without use of the conversational queries takes less than 2 min. Run-time with the conversational queries requires less than 3 min. The run-time is partly dependent on the speed of the printer.

#### PROGRAM PRINTOUT AND EQUATIONS

The complete program is presented in Fig. 2; the statement numbers, description, and symbols for each function are given in Fig. 3. The basic equations used are as follows (1):

Water vapor pressure ( $P_8$ ) =  $P_7/10^{Z_8}$  (760) (Ref. 2) (Step 6.2A or 8.1)

$$\text{where: } Z_8 = \left( \frac{X_8}{T_8} \right) \left[ \frac{A_7 + (B_7 \cdot X_8) + (C_7 \cdot X_8^3)}{1 + (D_7 \cdot X_8)} \right]$$

Expired  $O_2$  (Z) =  $B_2 + (M_2 \cdot F)$  (Step 10.2)

Expired  $CO_2$  (Y) =  $B_1 + (M_1 \cdot C)$  (Step 12.2)

Inspired  $CO_2$  (H) =  $B + (M \cdot H_1)$  (Step 16.1)

Inspired  $O_2$  (J) =  $B_2 + (M_2 \cdot J_1)$  (Step 17.2)

$\dot{V}_{EATPS}$  (W) =  $[(V_2 - V_1)/T] (V_3)$  (Step 20)

STPD FACTOR (U) =  $(P - P_8)/760(1 + 0.00367G)$  (Step 20)

$\dot{V}_{ESTPD}$  (V) =  $W \cdot U$  (Step 20)

$\dot{V}_{EBTPS}$  (O3) =  $[W - (R \cdot D)] \left[ \left( \frac{P - P_8}{P - 49.7} \right) \left( \frac{273 + 38}{273 + G} \right) \right]$  (Step 20)

where: 49.7 =  $H_2O$  vapor tension in expired air at  $T_b$  of 38°C.

$$R_E (R) = (Y\%) / [K[100-(Y+Z)]-Z] \quad (\text{Step 20})$$

$$\text{True } O_2 (X) = (Y\%) / R \quad (\text{Step 20})$$

$$O_2 \text{ uptake } (O) = (V \cdot X) / 100 \quad (\text{Step 20})$$

$$CO_2 \text{ production } (C4) = R \cdot O \quad (\text{Step 20})$$

The printout of a sample calculation is presented in Fig. 4.

A comparison was made between 22 measurements of oxygen uptake, that ranged from 1.79 to 4.50 l/min, calculated from the nomogram of Consolazio et al. (1) and from the computer program (Fig. 5). The mean ( $\pm$ SD) of the nomogram values was 3.44  $\pm$ 0.89 l/min versus 3.42  $\pm$ 0.88 for the program data. The correlation coefficient was 1.00.

#### REFERENCES

1. Consolazio, C. F., R. E. Johnson, and L. J. Pecora. Physiological Measurements of Metabolic Functions in Man. N.Y.: McGraw-Hill, 1963, pp. 1-11.
2. Keenan, J. H., and F. G. Keyes. Thermodynamic Properties of Steam. N.Y.: Wiley, 1963, p. 14.




FIG. 1. Flow diagram of program.

```

5      !THIS PROGRAM CALCULATES OXYGEN CONSUMPTION
7  PRINT
10 A9="NO"
11 B=-0.0143!87 !INTERCEPT FOR INSPIRED CO2 CURVE. SEE STATEMENT 830"
12 M= 0.00850467 !SLOPE FOR INSPIRED CO2 CURVE.
13 !
14 B1=-0.01066 !INTERCEPT FOR EXPIRED CO2 CURVE. SEE STATEMENT 670
15 M1= 0.03494 !SLOPE FOR EXPIRED CO2 CURVE.
16 !
17 B2=-0.002167 !INTERCEPT FOR INSPIRED AND EXPIRED O2 CURVE
18 M2= 0.02501 !SLOPE FOR INSPIRED AND EXPIRED O2 CURVE
20 PRINT "DO YOU NEED INSTRUCTIONS? TYPE YES OR NO: ":
25 INPUT IN IMAGE "#":I$
30 IF I$=A9 THEN 100 ELSE 35
35 PRINT
40 PRINT"THIS PROGRAM CALCULATES OXYGEN CONSUMPTION. IT CAN BE USED"
41 PRINT"WHEN EITHER OF TWO TYPES OF O2/CO2 ANALYZERS ARE AVAILABLE."
42 PRINT"THEY ARE: 1) DIRECT READOUT ANALYZERS (READINGS IN PERCENT OF"
43 PRINT"GAS COMPOSITION), AND 2) INDIRECT READOUT ANALYZERS (WHICH"
44 PRINT"REQUIRE USE OF A CALIBRATION CURVE)."
45 PRINT"THE SLOPES AND INTERCEPTS OF THE CALIBRATION CURVES MUST BE"
46 PRINT"ENTERED INTO THIS PROGRAM AT STATEMENT NUMBERS 11 TO 18."
47 PRINT
60 PRINT "WHEN REQUESTED TO TYPE (NO), DO NOT TYPE THE ENCLOSING (>)"
62 PRINT
64 PRINT "TO INTERRUPT THE EXECUTION OF THIS PROGRAM AT ANY POINT,"
66 PRINT "DEPRESS THE ALT MODE/ESCAPE KEY SEVERAL TIMES."
70 PRINT"THE PRIMARY EQUATIONS USED IN THIS PROGRAM PLUS THEIR"
71 PRINT"ASSOCIATED MNEMONICS CAN BE FOUND IN STATEMENTS 580,"
72 PRINT"1000 THROUGH 1090, AND 1325."
75 PRINT
80 PRINT"DATA CAN BE ENTERED IN TWO WAYS: 1) NON-CONVERSATIONALLY AS A"
81 PRINT"STRING OF VARIABLES IN A SPECIFIED FORMAT, OR 2)CONVER-"
82 PRINT"SATIONALLY, IN RESPONSE TO COMPUTER-GENERATED QUESTIONS."
83 PRINT"(CONVERSATIONAL? YES OR NO) GIVES YOU THE CHOICE OF HOW YOU"
84 PRINT"WANT TO INPUT THE DATA."
85 PRINT
100 PRINT
120 PRINT"CONVERSATIONAL? TYPE YES OR NO: ":
122 INPUT IN IMAGE "#":C9
123 PRINT
124 IF C9=A9 THEN 126 ELSE 200
126 PRINT "DO YOU NEED THE INPUT FORMAT? TYPE YES OR NO: ":
127 INPUT IN IMAGE "#":F8
128 IF F8=A9 THEN 158 ELSE 129
129 PRINT
130 PRINT"INSTRUCTIONS FOR ENTERING DATA ARE:"
131 PRINT
132 PRINT"      1) SEPARATE EACH VARIABLE WITH A COMMA"
133 PRINT"      2) PUSH THE LINE FEED KEY IF THE RIGHT-HAND EDGE OF THE"
134 PRINT"         PAPER IS REACHED BEFORE ALL VARIABLES ARE ENTERED"
135 PRINT"      3) TYPE (NO) IF BODY WT., ERGOMETER LOAD, OR TREADMILL"
136 PRINT"         SPEED ARE NOT MEASURED."
138 PRINT
139 PRINT "THE FORMAT FOR ENTERING DATA IS AS FOLLOWS:"
140 PRINT
141 PRINT"      COLUMN 1      COLUMN 2"
143 PRINT" D = DATE (ENCLOSE IN QUOTES) V1 = VOL. METER READ. INITIAL (L)"
144 PRINT" S$ = SUBJECT'S NAME      V3 = VOL. METER CORRECTION"
146 PRINT" W$ = BODY WT. (KG)      D$ = RESP. VALVE DEAD SPACE (L)"
148 PRINT" E$ = ERGOMETER LOAD (WATTS) G = GAS TEMPERATURE (CELSIUS)"
150 PRINT" T$ = TREADMILL SPEED (MPH) P = BAROMETRIC PRESS. (MM HG)"
152 PRINT" S = TIME, GAS COLLECT. (SEC) R$ = RESPIRATORY RATE (BR/MIN)"
154 PRINT" V2 = VOL. METER READ. FINAL (L)"
158 PRINT
160 PRINT"ENTER THE COLUMN 1 VARIABLES FIRST"
162 INPUT D,S$,W$,E$,T$,S,V2,V1,V3,D$,G,P,R$
164 PRINT

```

ORIGINAL PAGE IS  
OF POOR QUALITY

FIG. 2. Printout of complete program.



```

166 PRINT
170 PRINT "DATE =" ; D
171 PRINT
172 PRINT "SUBJECT =" ; S$
173 PRINT
174 PRINT "BODY WEIGHT (KG)          W$= " ; W$
175 PRINT "ERGOMETER LOAD (WATTS)    E$= " ; E$
176 PRINT "TREADMILL SPEED (MPH)    T$= " ; T$
177 PRINT
178 PRINT "TIME, GAS COLLECTION (SECONDS)    S= " ; S ; "SECONDS"
179 PRINT "RESP. GAS METER READING, FINAL    V2= " ; V2 ; "LITERS"
180 PRINT "RESP. GAS METER READING, INITIAL  V1= " ; V1 ; "LITERS"
181 PRINT "RESP. GAS METER CORRECTION        V3= " ; V3 ; "LITERS"
182 PRINT "RESP. VALVE DEAD SPACE           D$= " ; D$ ; "LITERS"
183 PRINT "GAS TEMPERATURE                 G= " ; G ; "CELSIUS"
185 PRINT "BAROMETRIC PRESSURE             P= " ; P ; "MM HG"
186 PRINT "RESPIRATION RATE               R$= " ; R$ ; "BR/MIN"
188 GO TO 500
200 PRINT "DATE (FORMAT = MONTH DAY, YEAR):";
210 INPUT IN IMAGE "##":D
215 PRINT
220 PRINT "SUBJECT'S NAME:";
230 INPUT IN IMAGE "##":S$
250 PRINT
260 PRINT "BODY WT. (KG) (TYPE <NO> IF NOT MEASURED) W$= " ;
270 INPUT IN IMAGE "##":W$
280 PRINT "ERGOMETER LOAD, (WATTS) (TYPE <NO> IF NOT MEASURED): " ;
290 INPUT IN IMAGE "##":E$
300 PRINT "TREADMILL SPEED, (MPH) (TYPE <NO> IF NOT MEASURED): " ;
310 INPUT IN IMAGE "##":T$
320 PRINT
330 PRINT "TIME, GAS COLLECTION (SEC.)          S= " ;
340 INPUT IN IMAGE "##":S
350 PRINT "RESP. GAS METER READ., FINAL (LITERS) V2= " ;
360 INPUT IN IMAGE "##":V2
370 PRINT "RESP. GAS METER READ., INITIAL (LITERS) V1= " ;
380 INPUT IN IMAGE "##":V1
390 PRINT "RESP. GAS METER CORRECTION          V3= " ;
400 INPUT IN IMAGE "##":V3
410 PRINT "RESP. VALVE DEAD SPACE              D$= " ;
420 INPUT IN IMAGE "##":D$
430 PRINT "GAS TEMPERATURE (CELSIUS)         G= " ;
440 INPUT IN IMAGE "##":G
460 PRINT "BAROMETRIC PRESSURE (MM HG)        P= " ;
470 INPUT IN IMAGE "##":P
480 PRINT "RESPIRATORY RATE (BR/MIN)          R$= " ;
490 INPUT IN IMAGE "##":R$
500 A7=3.2437814
510 B7=5.86826E-03
520 C7=1.1702379E-08
530 D7=2.1878462E-03
540 P7=218.167
550 T8=G+273.16
560 X8=647.27-T8
570 Z8=(X8/T8)*((A7+B7*X8+C7*X8+3)/(1+D7*X8))
580 P8=(P7/10:Z8)*760 !P8=PH2O SATURATED AT DBT
600 PRINT "WATER VAPOR PRESSURE                P8=" ; P8 ; " MM HG"
605 PRINT
630 PRINT "EXPIRED CO2, AS %, FROM DIRECT READOUT ANALYZER"
631 PRINT " (TYPE <NO> IF NOT MEASURED)          FCO2 Y= " ;
650 INPUT IN IMAGE "##":Y
660 IF Y=A9 THEN 670 ELSE 690
670 PRINT "EXPIRED CO2 (READING FROM INDIRECT READOUT ANALYZER) C= " ;
680 INPUT IN IMAGE "##":C
685 Y=B1+(M1*C) !FCO2 (%)
690 PRINT "EXPIRED O2, AS %, FROM DIRECT READOUT ANALYZER"
691 PRINT " (TYPE <NO> IF NOT MEASURED)          FO2 Z= " ;

```

Fig. 2.- Continued

```

700 INPUT IN IMAGE "#":Z
710 IF Z=A9 THEN 720 ELSE 745
720 PRINT "EXPIRED O2 (READING FROM INDIRECT READOUT ANALYZER) F= ":
730 INPUT IN IMAGE "#":F
740 Z=B2+(M2*F) !F02 (%)
745 PRINT
750 PRINT "WERE INSPIRED CO2 AND O2 MEASURED? TYPE YES OR NO: ":
755 INPUT IN IMAGE "#":N
757 PRINT
760 IF N=A9 THEN 770 ELSE 800
770 H=0.03
780 J=20.93
795 GO TO 919
800 PRINT "INSPIRED CO2, AS %, FROM DIRECT READOUT ANALYZER"
801 PRINT " (TYPE (NO) IF NOT MEASURED) H= ":
810 INPUT IN IMAGE "#":H
820 IF H=A9 THEN 830 ELSE 860
830 PRINT "INSPIRED CO2 (READING FROM INDIRECT READOUT ANALYZER) H1= ":
840 INPUT IN IMAGE "#":H1
850 H=B+(M*H1)
860 PRINT "INSPIRED O2, AS %, FROM DIRECT READOUT ANALYZER"
861 PRINT " (TYPE (NO) IF NOT MEASURED) J= ":
870 INPUT IN IMAGE "#":J
880 IF J=A9 THEN 890 ELSE 918
890 PRINT "INSPIRED O2 (READING FROM INDIRECT READOUT ANALYZER) J1= ":
900 INPUT IN IMAGE "#":J1
910 J=B2+(M2*J1)
918 PRINT
919 Q="XXXXXXXXXXXXXXXXX %%.%% %"
920 PRINT IN IMAGE Q:"INSPIRED CO2 H=":H:"%"
930 PRINT "INSPIRED O2 J=":J:"%"
940 PRINT "EXPIRED CO2 Y=":Y:"%"
950 PRINT "EXPIRED O2 Z=":Z:"%"
1000 K=J/(100-(J+H)) !INSPIRED O2/N2
1005 T=S/60
1010 Y$=Y-H !EXPIRED CO2 MINUS INSPIRED CO2
1020 W=(V2-V1)/T*V3 !VE ATPS (L/MIN)
1030 U=(P-P8)/(760*(1+0.00367*G)) !STPD FACTOR
1040 V=W*U !VE STPD (L/MIN)
1050 O3=(W-(R$+D$))*(((P-P8)/(P-49.7))*(273+38)/(273+G)) !VE BTPS (L/MIN)
1060 R=Y$/(100-(Y+Z))*K-Z !RE
1070 X=Y$/R !TRUE O2 (%)
1080 O=(V*X)/100 !OXYGEN UPTAKE (L/MIN)
1090 C4=R*O !CO2 PRODUCTION (L/MIN)
1120 PRINT
1200 PRINT "TIME =":T;"MINUTES"
1210 PRINT "VE ATPS =":W;"L/MIN"
1220 PRINT "STPD FACTOR =":U
1230 PRINT "VE STPD =":V;"L/MIN"
1240 PRINT "VE BTPS =":O3;"L/MIN"
1250 PRINT "FCO2 MINUS INSPIRED CO2 =":Y$:"%"
1260 PRINT "FO2 =":Z:"%"
1270 PRINT "RE =":R
1280 PRINT "TRUE O2 =":X:"%"
1290 PRINT "CO2 PRODUCTION =":C4;"LITERS PER MINUTE"
1300 PRINT
1310 PRINT "OXYGEN UPTAKE =":O;"LITERS PER MINUTE"
1320 IF W$=A9 THEN 1398
1325 O1=(O*1000)/W$ !OXYGEN UPTAKE (ML/KG/MIN)
1330 PRINT "OXYGEN UPTAKE =":O1;"ML/KG/MIN"
1398 PRINT
1400 PRINT "ARE YOU FINISHED? TYPE YES OR NO: ":
1410 INPUT IN IMAGE "#":F9
1415 PRINT
1420 IF F9=A9 THEN 120 ELSE 1440
1440 END

```

ORIGINAL PAGE IS  
OF POOR QUALITY

Fig. 2.- Concluded

# SYMBOLS AND CONSTANTS USED IN O<sub>2</sub>/UPTAKE PROGRAM

Symbol		Statement No.
A7	Water Vapor Pressure Constant (3.2437814)	500
A9	"No"	10
B	Intercept for Inspired CO <sub>2</sub> Curve	11
B1	Intercept for Expired CO <sub>2</sub> Curve	14
B2	Intercept for Inspired & Expired O <sub>2</sub> Curve	17
E7	Water Vapor Pressure Constant (5.86826X10 <sup>-3</sup> )	510
C	Expired CO <sub>2</sub> Reading from Indirect Readout Analyzer	670, 680
C4	CO <sub>2</sub> Production (L/Min)	1090
C7	Water Vapor Pressure Constant (1.1702379X10 <sup>-8</sup> )	520
C9	Prompt for Conversational Mode of Input	120, 122
D	Date	162, 200, 210
D7	Water Vapor Pressure Constant (2.1878462X10 <sup>-3</sup> )	530
D5	Respiratory Valve Dead Space (LITERS)	162, 410, 420
E5	Ergometer Load (Watts)	162, 280, 290
F	Expired O <sub>2</sub> Reading from Indirect Readout Analyzer	720, 730
F8	Branch in Program for Format Instructions	126, 127
F9	Branch in Program for Exiting	1400, 1410
G	Gas Temperature (CELSIUS)	162, 430, 440
H	Inspired CO <sub>2</sub> , Constant or Direct Reading (%)	770, 801, 810
H1	Inspired CO <sub>2</sub> Reading from Indirect Readout Analyzer	830, 840
I\$	Branch in Program for General Instructions	20, 25
J	Inspired O <sub>2</sub> , Constant or Direct Reading	780, 861, 870
J1	Inspired O <sub>2</sub> Reading from Indirect Readout Analyzer	890, 900
K	Inspired O <sub>2</sub> /N <sub>2</sub> (%)	1000
M	Slope for Inspired CO <sub>2</sub> Curve	12
M1	Slope for Expired CO <sub>2</sub> Curve	15
M2	Slope for Inspired and Expired O <sub>2</sub> Curve	18
N	Branch in Program for Inspired Air Measurements	750, 755
O	Oxygen Uptake (L/Min)	1080
O1	Oxygen Uptake (ML/KG/Min)	1325
O3	VE BTPS (L/MIN)	1050
P	Barometric Pressure (TORR)	162, 460, 470
P7	Water Vapor Pressure Constant (218.167 INT. ATM)	540
P8	Water Vapor Pressure (INT. ATMOSPHERES)	580
Q	Print in Image Statement	913
R	RE	1060
R\$	Respiratory Rate (BREATHS/MIN)	162, 480, 490
S	Time (Seconds)	162, 330, 340
S\$	Subject's Name	162, 220, 230
T	Time (Minutes)	1005
T8	Gas Temperature, KELVIN (G, °C + 273.16)	550
T\$	Treadmill Speed	162, 300, 310
U	STPD Factor	1030
V	VE STPD (L/MIN)	1040
V1	Respiratory Gas Meter Reading, Initial (LITERS)	162, 370, 380
V2	Respiratory Gas Meter Reading, Final (LITERS)	162, 350, 360
V3	Respiratory Gas Meter Correction (LITERS)	162, 390, 400
W	VE ATPS (L/MIN)	1020
W\$	Body Weight (kg)	162, 260, 270
X	True O <sub>2</sub> (%)	1070
X8	647.27-T8	560
Y	Expired CO <sub>2</sub> from Direct Readout Analyzer (%)	631, 650
Y\$	Net CO <sub>2</sub> (%)	1010
Z	Expired O <sub>2</sub> from Direct Readout Analyzer (%)	691, 700
Z8	LOG <sub>10</sub> P7/P8	570

FIG. 3. Symbols and constants used in the program.

ENTER THE COLUMN 1 VARIABLES FIRST

? "3/30/77",SUBJECT,66.42,175,NO,60,76.5,0,1.066,0,27.3,765.3,40

DATE = 3/30/77

SUBJECT =

BODY WEIGHT (KG) W\$= 66.42

ERGOMETER LOAD (WATTS) E\$= 175

TREADMILL SPEED (MPH) T\$= NO

TIME, GAS COLLECTION (SECONDS)	S= 60 SECONDS
RESP. GAS METER READING, FINAL	V2= 76.5 LITERS
RESP. GAS METER READING, INITIAL	V1= 0 LITERS
RESP. GAS METER CORRECTION	V3= 1.066 LITERS
RESP. VALVE DEAD SPACE	D\$= 0 LITERS
GAS TEMPERATURE	G= 27.3 CELSIUS
BAROMETRIC PRESSURE	P= 765.3 MM HG
RESPIRATION RATE	R\$= 40 BR/MIN
WATER VAPOR PRESSURE	P8= 27.208617 MM HG

EXPIRED CO2, AS %, FROM DIRECT READOUT ANALYZER

(TYPE <NO> IF NOT MEASURED) FC02 Y= 3.99

EXPIRED O2, AS %, FROM DIRECT READOUT ANALYZER

(TYPE <NO> IF NOT MEASURED) FO2 Z= NO

EXPIRED O2 (READING FROM INDIRECT READOUT ANALYZER) F= 657

WERE INSPIRED CO2 AND O2 MEASURED? TYPE YES OR NO: NO

INSPIRED CO2 H= .030 %

INSPIRED O2 J= 20.93%

EXPIRED CO2 Y= 3.99%

EXPIRED O2 Z= 16.429403%

TIME	=	1 MINUTES
VE ATPS	=	81.549 L/MIN
STPD FACTOR	=	.88273116
VE STPD	=	71.985843 L/MIN
VE BTPS	=	87.109095 L/MIN
FC02 MINUS INSPIRED CO2	=	3.96%
FO2	=	16.429403%
RE	=	.85275936
TRUE O2	=	4.6437485%
CO2 PRODUCTION	=	2.8506394 LITERS PER MINUTE

OXYGEN UPTAKE = 3.3428415 LITERS PER MINUTE

OXYGEN UPTAKE = 50.328839 ML/KG/MIN

ARE YOU FINISHED? TYPE YES OR NO:

FIG. 4. Sample printout of O<sub>2</sub> uptake computation.




FIG. 5. Comparison of oxygen uptake calculated with the computer program and the nomograph method.